
ABOUT THIS BOOK

Serena™ StarTool® FDM StarBat Option is a part of the Serena™ StarTool® family of 
products; it is a batch program that is normally used for repetitive, bulk data 
processing tasks. This manual contains detailed information on the use of the StarBat.

This document describes Serena™ StarTool® FDM StarBat Option, Version 7.4, a 
product of Serena™ Software Inc. This product, previously called, StarBat for 
StarTool®, has recently undergone a name change. The new name more clearly 
reflects the role of the product within the Serena™ family of products.

The product may also be referred to as File and Data Manager or FDM.

Note that the old product name, as well as variants of the new product name, are used 
in this document.

Note:

StarTool, with its StarBat option, is designed to provide a superior file and data 
management solution for application developers and system programmers at a more 
attractive price than major competitors. StarBat is not a drop in replacement for File-
AID/Batch MVS and there is no guarantee that StarBat MVS has all the functionality of 
File-AID/Batch MVS. StarTool provides many features and functionality not found in 
File-AID MVS.

File-AID® is a registered trademark of Compuware Corporation.

OBJECTIVES
Use this book to learn the functionality available in the StarBat Option of StarTool 
FDM.



8  

SUPPORT
Serena Software provides technical support on the Internet through the Serena 
eSupport self-service Web site. To access the eSupport site, go to 
http://support.serena.com and login with your ID and password to see the eSupport 
Customer Portal. From there, you can:

• Report new issues.

• Search our problem-tracking system for information about existing problems.

• View a knowledge base of frequently asked questions and helpful product hints.

• Query the call tracking database to obtain the current status of an open issue.

• Access our FTP server to download product fixes and documentation in PDF 
format.

• Subscribe to one of our mailing list servers (LISTSERVE) to receive the latest 
product information by e-mail.

If you need a password or ID, please e-mail or call the appropriate support department 
for your country.

United States, Canada United Kingdom, Germany, 
Austria, France, Switzerland, 
BENELUX

Support 
Hours

Monday through Friday
5:00 A.M. to 5:00 P.M. 
(Pacific Time)

Monday through Friday 
7:30 A.M. to 6:30 P.M. 
(United Kingdom Time)

E-mail support@serena.com support@serena.com

URL http://support.serena.com http://support.serena.com

Phone 877.696.1850 44 1494 765888

Facsimile 650.522.6698 44 1494 766888

For support in all other countries, please contact your local distributor. 

mailto:support@serena.com
mailto:support@serena.com
http://support.serena.com
http://support.serena.com
http://support.serena.com
http://support.serena.com

INTRODUCTION 1
StarBat is an MVS batch program using standard MVS JCL to manipulate data files. 
StarBat supports data in character, hexadecimal, packed, and binary formats on disk 
or on tape.

Using StarBat, you can copy, update, or print an entire file or specific records within 
the file that satisfy certain conditions. You can also use StarBat to process numeric 
data, modify dates, and convert date formats.

In this introduction, we briefly describe the main functions that StarBat provides. We 
introduce the major functions and explain how to combine the functions and 
parameters into a job to modify a data file. 

FEATURES
There are two main uses for StarBat. 

• Manipulate members of a data set
Different StarBat functions can copy, print and update members of a data set. 
These functions are COPYMBR, PRINTMBR, PRINTCHRMBR, PRINTHEXMBR, 
and UPDATEMBR.

• Manipulate records in a member or data set
This is by far the most common use of StarBat. This allows for the copying, 
deleting, and inserting of records. Additionally, records can be printed as text, 
hexadecimal, or both. The data in the records can be totaled and reformatted in a 
variety of ways. The functions used here are COPYALL, COPYREC, COPYREV, 
COPYSOME, EXCLUDEREC, MULTICOPY, PRINT, PRINTALL, PRINTCHR, 
PRINTCHRALL, PRINTCHRREV, PRINTHEX, PRINHEXALL, PRINTHEXREV, 
PRINTREV, SKIP, SKIPREV, TOTAL, UPDATEALL, and UPDATEREC.

If we look at the functions themselves, they fall into five categories.

• Copying data
There are many occasions when it is required to create a subset of a data file and, 
by combining the various copy functions, the exclude function and the conditional 
parameters, it is possible to develop sophisticated jobs to manipulate records to 
create new data files.

• Deleting data
Whether it’s deleting bad records or dropping records because they are no longer 
required, StarBat has the ability to do this effortlessly. Again, in combination with 
conditional parameters, it is possible to edit a file in batch mode without having to 
create a special purpose program to do it.

• Inserting data
In addition to copying and deleting records, StarBat can insert new records if 
required. This allows new record types to be added, for example, to a test data file 
so that new functionality in a program can be tested.

• Printing data
StarBat allows for data sets and members to be printed in text mode, which is 
suitable for source files for example, and in hexadecimal mode, which is more 
suited to data files. This is invaluable in testing a program because it clearly 
shows the exact contents of the files.

• Changing data
One of the most powerful features of StarBat is its ability to modify the content of a 
record. Using commands similar to ISPF edit commands, it is possible to change 
whole files or individual records according to some defined pattern. These 
changes include the ability to increase the size of a field, as might be needed with 
converting a date format from MMDDYY to YYYYMMDD.

TYPES OF DATA SET PROCESSING
StarBat can process disk files with partitioned (PDS), partitioned extended (PDSE), 
sequential, VSAM, or direct organization; it can process tape files with sequential 
organization. Any record format is acceptable; StarBat assumes that any partitioned 
data set with record format U is a load library and that other libraries are usually called 
source libraries.

NEW FEATURES
Some of StarBat’s new features are:

• Records can be copied in whole or in part from any data set of any type to any 
other data set of the same or a different type

• Records can be printed in text or vertical hexadecimal

• Data items within records can be replaced with larger or smaller data items

• An entire data set can be processed at one time, or records can be processed 
selectively

• Totals can be created for particular data within the data set

• Data or date fields can be warped

Note: you will be able to use StarTool FDM and StarBat Option features only if you are 
licensed to use both products. Please contact SERENA Software sales if you want to 
arrange a trial of the other product.

SOFTWARE ENVIRONMENT
StarBat runs under the major IBM operating systems:

• MVS/ESA (any release)

• OS/390 (any release)

• z/OS (any release)

In addition, the following environments should be available:

• ISPF and ISPF/PDF (Version 4.0 or above)

• TSO/E (any release or any version)

STARBAT INITIALIZATION
When StarBat is invoked, license information is normally displayed as shown below to 
identify the program, release number and release date.

PDS100I StarBat/Both -- Version 7.4.0  2000.350

Proprietary software product of SERENA Software 
Phone (877)696-1850 OR FAX (650)522-1776
LICENSED TO: your corporate name/trial offer expires ...
             your city, state, zip/agent to contact for license ...
All other rights reserved - use of this software
product by unauthorized persons is prohibited.

PRODUCT OVERVIEW 2
StarBat is a program designed to perform a variety of data manipulation tasks in a 
background environment.

You can use StarBat to copy selected records or portions of records from one data set 
type to another, print data in different formats, add or enlarge data fields in records, 
warp date fields, and process data set members selectively.

StarBat is a MVS batch program that is controlled with standard MVS JCL. You 
provide control statements to specify the functions you want performed in your JCL or 
in a data set pointed to by the SYSIN DD statement.

For StarBat, functions are defined to process data sets to your specifications. 
Parameters specified on these function statements limit the parts of the data set to be 
processed and control secondary processes.

FUNCTIONS AND MODIFIERS
Following is a brief description of each supported function.

Function Short Description

COPYALL CA Copies all records of a data set

COPYMBR CM Copies members conditionally based on contents

COPYREC CR Copies data and reports record totals

COPYREV CPR Copies data in reverse order

COPYSOME CS Copies selected records but applies all changes 
like COPYALL

EXCLUDEREC XR Eliminates unwanted records in a copy

MULTICOPY MC Copies data to one or more output data sets

PRINT P Prints records in alphanumeric format with record 
statistics

PRINTALL PA Prints all records of a data set

PRINTCHR PC Prints records in alphanumeric format

PRINTCHRALL PCA Prints all records of a data set

PRINTCHRMBR PCM Prints members conditionally based on contents

PRINTCHRREV PCR Prints records in alphanumeric format and reverse 
order

PRINTHEX PH Prints records in vertical hexadecimal format

PRINTHEXALL PHA Prints all records of a data set

PRINTHEXMBR PHM Prints members conditionally based on contents

PRINTHEXREV PHR Prints records in vertical hexadecimal format and 
reverse order

PRINTMBR PM Prints members conditionally based on contents

PRINTREV PRR Prints records in alphanumeric format and reverse 
order

SKIP S Moves the current record pointer forward

SKIPREV SKR Moves the current record pointer backward

TOTAL T Reads input records processing all parameter 
groups for SUM

UPDATEALL UA Updates all records of a data set

UPDATEMBR UM Updates members conditionally based on contents

UPDATEREC UR Updates records in place

Function Short Description

PARAMETER DESCRIPTIONS
Parameters are code words or keywords that control processing actions. 

Following is a brief description of each supported parameter.

Parameter Short Description

ABEND AB Controls EOJ processing when abnormal condition 
occurs

AND IF Creates a logical AND condition check (used with IF)

CHANGE C Changes only the first instance of data in a record

CHANGEALL CA Changes all occurrences of data in a record

COPYOVER CO Controls the replacement of identically named output 
members

EXCLUDEREC XR Controls the # of records to bypass in an 
EXCLUDEREC function

EXPAND EX Expands records at a specified location 

IF AND Selects records to process based on data contents

MAXRECIN MRI Controls the number of records to input

MAXRECOUT MRO Controls the maximum number of records to output

MEMBER M Specifies a member name to process in a PDS

MEMBERS MS Specifies a group of members to process in a PDS 
with a mask

MOVE MV Moves data into the record

NEWMBR NM Gives a new name to an output PDS member

NEWMBRS NMS Names multiple new members of an output PDS using 
a mask

OPTIONS OP Controls StarBat processing options

OR (none) Used with the IF parameter to indicate an OR condition

OVERALL OA Replaces all occurrences of data in a record with other 
data

OVERLAY OL Replaces the first instance of data in a record with new 
data

PADCHAR PAD Specifies a padding character for uninitialized parts of 
a record

PRINT P Prints records in alphanumeric format with record 
statistics

PRINTCHR PC Prints records in alphanumeric format

PRINTHEX PH Prints records in vertical hexadecimal format

PRINTLPI PL Specifies the number of lines per inch for print output 
pages

RBA (none) Processes VSAM data beginning at a relative byte 
address

RDW (none) Controls the inclusion of the record descriptor word

SELECT S Processes every nth record

STARTKEY SK Processes VSAM data beginning with a generic key

STOPIF ST Stops processing a function when a record satisfies a 
condition

SUM (none) Accumulates the contents of specified fields

WRITE W Writes a record to one or more output files

Parameter Short Description

DATA SETS 3
StarBat supports sequential, direct, VSAM, and partitioned data sets. ISAM data sets 
are not supported.

SEQUENTIAL DATA SETS
StarBat can create any type of sequential data set as output. Input concatenated data 
sets on unlike devices are supported but these data sets must have similar 
characteristics. When processing concatenated input data sets, StarBat treats all of 
these data sets as one logical input data set and no reports are provided on the 
individual data sets in the concatenation aside from a record count for each 
concatenated data set.

DIRECT DATA SETS
StarBat uses BSAM to read direct data sets. If RKP is greater than zero, you do not 
need to perform any special action to copy the data and the imbedded keys. However, 
if RKP is zero, you will need to use a MOVE parameter with a negative relative 
location to reference the key portion because the key is not also present in the data.

For example, if you have a record with a 20 byte key and a 80 byte data portion, use 
the following MOVE parameter to copy both the key and data portion of the record to 
an output record:

DD01  COPYREC   MOVE=(1,100,-20)

VSAM DATA SETS
StarBat supports RRDS, KSDS, and ESDS data sets. Linear VSAM data sets are 
supported but the CISIZE has to be 4K.

PARTITIONED DATA SETS
StarBat supports PDS and PDSEs. To process a specific member, use the MEMBER 
parameter. To process multiple members, either use multiple control statements with 
the MEMBER parameter (only one MEMBER parameter is allowed for each control 
statement), or use the MEMBERS parameter specifying a member mask. 
To process the entire data set, do not specify any MEMBER or MEMBERS parameter.

If the PDS and PDSEs are concatenated as input and if the MEMBER parameter is 
used, StarBat processes each data set in the order of concatenation and stops 
processing the function as soon as the member is found.

There are a few restrictions when processing load library members: 

• Load PDSE members cannot be used as output in a copy.

• The MULTICOPY function does not support scatter-loaded, overlay, or note-listed 
modules.

• The block size of the input data set must be equal to or less than the output data 
set.

• The main module associated with an alias member is always copied at the same 
time an alias is copied.

STARBAT JCL AND DDNAMES
A sample of JCL to execute StarBat follows. 

//StarBat1 JOB  (JOB CARD PARAMETERS)   
//A   EXEC PGM=StarBat,REGION=2048K
//STEPLIB  DD  DSN=STEPLIB.DATA.SET,DISP=SHR
//DD03     DD  DSN=SAMPLE.INPUT.DATA.SET,DISP=SHR
//DD03O    DD  DSN=SAMPLE.OUTPUT.DATA.SET,DISP=SHR
//SYSPRINT DD  SYSOUT=*
//SYSLIST  DD  SYSOUT=*
//SYSTOTAL DD  SYSOUT=*
//SYSIN    DD  *
DD03 COPYALL=(1,0,C'PGM=TESTPROG',C'PGM=PRODPROG'),
            PRINTCHR=100

The StarBat job looks like any other MVS job. The usual rules for EXEC and DD 
statements apply.

The EXEC statement tells the operating system the name of the program to be 
executed. A two-megabyte region is usually sufficient for StarBat.

A STEPLIB or JOBLIB statement is required unless StarBat is present in your 
system’s link list.

A DDxx statement is required to describe input data sets to StarBat. The xx value can 
be any number from 00 through 99 that matches a control statement name like DDxx. 
Note that input data sets may be concatenated.

A DDxxO statement is required for COPYREC or EXCLUDEREC functions. The xx 
value must match the xx value in the input data set defined on the corresponding 
DDxx JCL statement. If necessary, basic DCB information such as RECFM, LRECL, 
and BLKSIZE will be copied from the input data set to the output data set when the 
data set is opened. 

Additional output DD statements with any desired name may be specified for the 
MULTICOPY function; the WRITE parameter names the output DD statement and 
causes output records to be written when it is processed. 

StarBat primary output is directed to the SYSPRINT DD statement. This statement is 
optional; it will be dynamically allocated if it is not present. StarBat echoes control 
statements and error, status, and completion messages to this DD name. StarBat 
expects this to be a data set with standard attributes for a listing file namely: 
RECFM=FBA, LRECL=133, and BLKSIZE a multiple of 133. Note that an LRECL of 
80 through 132 will work but will truncate some print lines.

If you specify the PRINT or PRINTHEX options, the SYSLIST DD statement should be 
included. If it is omitted, it is dynamically allocated if needed. This data set has the 
same DCB attributes as the SYSPRINT data set. 

If you specify the SUM option, the SYSTOTAL DD statement should be included. If it 
is omitted, all SYSTOTAL output is re-directed to the SYSPRINT data set. This data 
set also has the same DCB attributes as the SYSPRINT data set. 

The SYSIN statement provides control statements for StarBat. Control cards must be 
in standard 80-character format and all 80 columns may be used. If no control cards 
are found, StarBat defaults to a COPYREC function for every pair of matching input 
and output data sets (//DDxx and //DDxxO).

CONTROL STATEMENTS 4
StarBat control statements consist of a data set identifier, a function name, and one or 
more parameters separated by commas. Comments are preceded by one or more 
blanks following a parameter.

In the following example, DD01 is the data set identifier. PRINTCHR is the function 
name, IF=(10,EQ,C’A’) is the parameter and the remainder of the statement is 
comments.

DD01 PRINTCHR IF=(10,EQ,C’A’)  prints records containing character ’A’ in column 10.

DATA SET IDENTIFIER
The data set identifier identifies which data set in the JCL DD statement is being 
processed. The format of the data set identifier is DDxx where xx is from 00 to 99. The 
number references the //DDxx DD statement, which specifies the input data set to be 
processed. DDxx must begin in column 1 of the control statement. If the function is 
COPY or EXCLUDEREC, DDxx is also used to reference the //DDxxO DD statement 
representing the output data set.

FUNCTION IDENTIFIER
The function identifier identifies the action that you want to perform on the data. It is 
always positioned after the data set identifier with one or more spaces.

PARAMETERS
Parameters are conditions defining the scope of the function; they are positioned after 
the function identifier and one or more spaces. Each parameter is followed by an 
equal sign and one or more elements containing a combination of start, length, 
operator, and data fields. There can be one or more parameters separated by 
commas.

Start
A start location indicates the location where data can be found in the input record. A 
start location may be coded in one of two ways: actual start location and relative start 
location.

Actual start location is the exact column number where the data in the record is 
positioned. The number can be from 1 to 32767, but it cannot be larger than the 
record size.

Relative start location indicates the position relative to the current location. StarBat 
supports both input relative start location and output relative start location. 

When an input record is first read by StarBat, the relative location is set to the 
beginning of the record. Parameters such as IF, CHANGE, and OVERLAY change 
this location value. A relative start location can be specified by placing a plus sign or a 
minus sign before a number. For example, to reference a location 10 bytes before the 
current location, use –10; to reference a location 5 bytes after the current location, use 
+5. Output relative location is supported when using the MOVE parameter with the 
COPY or MULTICOPY functions. When data is moved to an output location, the 
output relative location moves to the next available output position.

Operator
Operators are EQ, NE, GT, LT, GE, and LE. They are used to compare data with an 
input record.

StarBat also supports bit value comparisons using operators AO (all ones), AZ (all 
zeroes), NO (not ones), and MX (mixed ones and zeroes) with the IF, OR, CHANGE, 
or OVERLAY parameters.

Length
When the exact location of the compare data is unknown, the length element may be 
used in place of the operator element. In this case, an equal comparison is assumed. 

To scan the entire input record, use a length of 0. Otherwise, the length value must be 
at least one greater than the length of the compare data; also, the sum of the current 
input location and the length value must be less than or equal to the record length. 
Length cannot be greater than 255.

Note that the IF, OR, CHANGE and OVERLAY parameters are changed to scanning 
parameters when the length element is used. Therefore, they will change the relative 
start input position as previously mentioned.



23

Serena™ StarTool® FDM 7.4 StarBat Option Getting Started Guide

Data
StarBat supports character, hexadecimal, binary, and packed data. The data element 
can be used as compare data, as replacement data, or as literal data.

Character data may be enclosed in single or double quotes. When used as compare 
data, more than one value can be entered with commas in between, indicating an OR 
condition, as long as they are enclosed in single quotes. For example, IF=(10,EQ, 
C’TEST,PROD’) checks if the record contains the characters TEST or PROD in 
column 10. A duplication factor may be used to avoid coding repetitive data elements. 
For example, instead of coding 
C‘TEST,PROD,TEST,PROD,’, you could code 2C’TEST,PROD,’.

Alphanumeric data is case-sensitive and is not translated to upper case. To search for 
data in both upper and lower case, use the T character string. For example, to search 
for PROD, Prod, or proD, use T’Prod’.

Note that all character data must fit into a single control statement because 
continuations are not supported.

Packed data can be entered with any valid length that fits on a single control 
statement (continuations are not supported). You must enclose all numeric digits in 
single quotes. StarBat calculates the data length using the data entered. You may 
enter a plus or minus before each string of numbers; an unsigned packed number is 
considered positive. For example, P'12' and P'+12' are equivalent.

Binary data can be entered with a beginning H character for halfword binary numbers 
or an F character for fullword binary numbers. You must enclose all numeric digits in 
single quotes. You may enter a plus or minus before each string of numbers; an 
unsigned binary number is considered positive. For example, F'12' and H'-12'.

Hexadecimal data can be entered with any length that fits on a single control 
statement (continuations are not supported). You must enclose all hexadecimal digits 
in single quotes. StarBat calculates the data length using the data entered and it must 
always have an even number of digits. You may also use a numeric duplication factor 
to avoid coding repetitive data elements. For example, 3X'003F' and 
X'003F003F003F' are equivalent.

Comments
Comments may be coded after a parameter or a parameter with a comma as long as 
there are one or more blanks between them. A comment may also be coded as a 
standalone statement provided that the statement begins with an *. In this case, the 
comment will also be output in the SYSTOTAL data set.

Three comments are coded in the following example: 

DD03 COPYREC MAXRECIN=10, COMMENT#1
            COPYOVER=YES COMMENT#2
* COMMENT#3

Following is a brief description of each supported function.

Function Short Description

COPYALL CA Copies all records of a data set

COPYMBR CM Copies members conditionally based on contents

COPYREC CR Copies data and reports records copied

COPYREV CPR Copies data in reverse order

COPYSOME CS Copies selected records but applies all changes 
like COPYALL

EXCLUDEREC XR Eliminates unwanted records in a copy

MULTICOPY MC Copies data to one or more output data sets

PRINT P Prints records in alphanumeric format with record 
statistics

PRINTALL PA Prints all records of a data set

PRINTCHR PC Prints records in alphanumeric format

PRINTCHRALL PCA Prints all records of a data set

PRINTCHRMBR PCM Prints members conditionally based on contents

PRINTCHRREV PCR Prints records in alphanumeric format and reverse 
order

PRINTHEX PH Prints records in vertical hexadecimal format

PRINTHEXALL PHA Prints all records of a data set

PRINTHEXMBR PHM Prints members conditionally based on contents

StarBat function names can be abbreviated as shown in the title lines in the following 
pages.

COPYREC, COPYALL, COPYMBR, AND 
COPYSOME

Copies data and reports records output. If it is required to copy records from one data 
set into another, the COPYREC (or CR) function is used. With the addition of 
qualifiers it is possible to control the records selected for copying. Additionally, in the 
process of copying, the data can be modified by expanding or contracting fields and 
changing the data within fields.

Use COPYMBR (or CM) to copy members based on their contents and use 
COPYALL (or CA) to process multiple conditional updates 
(IF OVERLAY, IF CHANGE, IF MOVE, IF EXPAND, IF WARP) while copying all 
records. Use the COPYSOME (or CS) function to process multiple conditional 
updates like COPYALL while copying selected records like COPYREC.

PRINTHEXREV PHR Prints records in vertical hexadecimal format and 
reverse order

PRINTMBR PM Prints members conditionally based on contents

PRINTREV PRR Prints records in alphanumeric format and reverse 
order

SKIP S Moves the current record pointer forward

SKIPREV SKR Moves the current record pointer backward

TOTAL T Reads input records processing all parameter 
groups for SUM

UPDATEALL UA Updates all records of a data set

UPDATEMBR UM Updates members conditionally based on contents

UPDATEREC UR Updates records in place

Function Short Description

The following example will copy up to 1000 records that contain USA anywhere after 
position 256:

DD01 COPYREC IF=(256,0,C'USA'),MAXRECOUT=1000

The following example will copy the input data set and replace the first USA in the first 
three positions with DOM:

DD01 COPYREC OVERLAY=(1,3,C'USA',C'DOM')

The following example will copy every other record that contains USA in the first three 
positions:

DD01 COPYREC SELECT=2,IF=(1,3,C'USA')

The following example will copy all members whose names begin with STAR 
containing USA anywhere in the member:

DD01 COPYMBR MEMBERS=STAR-,IF=(1,0,C'USA')

The following example will copy records containing an 8 or 9 and warp each record 
with YYDDD or YYMMDD format. Note that if a record has both formats, only the 
YYDDD format will be warped.

DD01 COPYREC   IF=(12,EQ,C'8'),
               WARP=(16,C,YYDDD,ADD=5Y),
               IF=(22,EQ,C'9),
               WARP=(26,C,YYMMDD,ADD=5Y)

The following example will copy all records and warp all 8 and 9 dates:

DD01 COPYALL   IF=(12,EQ,C'8'),
               WARP=(16,C,YYDDD,ADD=5Y),
               IF=(22,EQ,C'9),
               WARP=(26,C,YYMMDD,ADD=5Y)

The following example will copy records containing an 8 or 9 and warp all dates on 
these records:

DD01 COPYSOME  IF=(12,EQ,C'8'),
               WARP=(16,C,YYDDD,ADD=5Y),
               IF=(22,EQ,C'9),
               WARP=(26,C,YYMMDD,ADD=5Y)

COPYREV
The COPYREV or CR function copies records from a sequential or VSAM data set in 
reverse order. This means that the first record copied will be the last record from the 
input data set followed by other records from the input data set moving toward the 
front of the data set.

The following example will copy records except those containing RED anywhere in 
reverse order:

DD01 COPYREV IF=(1,0,C'RED')

The following example will copy up to 1000 records that contain USA anywhere after 
position 256 in reverse order:

DD01 COPYREV IF=(256,0,C'USA'),MAXRECOUT=1000

EXCLUDEREC
The EXCLUDEREC or XR function excludes (or eliminates) records during a copy; 
EXCLUDEREC selects records that would be bypassed by COPYREC.

The following example will copy all records except those containing RED anywhere:

DD01 EXCLUDEREC IF=(1,0,C'RED')

The following example will copy all records except the first 100 containing RED 
anywhere:

DD01 EXCLUDEREC IF=(1,0,C'RED'),EXCLUDEREC=100

The following example will copy the input data set and drop any record that contains 
an A in position 1 or contains a value higher than CAA in location 112. Processing 
stops after copying 20 records:

DD01 EXCLUDEREC IF=(1,EQ,C'A'),OR=(112,GT,C'AA'),MAXRECOUT=20

MULTICOPY
The MULTICOPY or MC function copies data to one or more output data sets. With 
the MULTICOPY function, you can direct output records to several output data sets 
and add, delete, or modify records being copied to the output data sets at the same 
time.



29

Serena™ StarTool® FDM 7.4 StarBat Option Getting Started Guide

In order to control the destination of the records, the WRITE parameter is used to 
specify the output DDNAME.

Output records for MULTICOPY are created piece by piece with MOVE or by copying 
the input record. If the MOVE parameter is used, the output area is initialized to the 
PADCHAR value and not reset between different WRITE parameters. 

There are several restrictions on the data that can be output with a MULTICOPY 
function:

Output to load members is supported; however, JCL indicating PDS(MEMBER) 
notation is not supported. Instead, use only the PDS name in the JCL and indicate the 
member name with the MEMBER, MEMBERS, NEWMBR, or NEWMBRS 
parameters. 

Overlay, scatter-loaded, and note-listed load members are not supported.

The following example shows how to create two identical output data sets in //OUT1 
and //OUT2:

DD01 MULTICOPY WRITE=(OUT1,OUT2)

The following example will copy a file, repeat any record with a C'5' and change it to a 
C'6':

DD01 MULTICOPY WRITE=OUT1,IF=(20,EQ,C'5'),MOVE=(1,0,1),
            MOVE=(20,C'6'),WRITE=OUT1

The following example will add a STEPLIB JCL statement after any EXEC statement:

DD01 MULTICOPY WRITE=NEWJCL,IF=(1,20,C' EXEC '),MOVE=(1,80C' '),
            MOVE=(1,C'//STEPLIB  DD  DISP=SHR,'),
            MOVE=(+0,C'DSN=MYHILEV.MYMIDLEV.MYLOWLEV'),WRITE=NEWJCL

PRINT, PRINTALL, PRINTMBR, AND 
PRINTREV

Print (or P) prints a data set or a portion of a data set in alphanumeric format with 
identifying information such as record number, RBA, and record length. A column 
scale is normally printed with each record; however, if you request 
OPTIONS=SHORT, the column scale will be produced only after each page header.

PRINTMBR (or PM) prints members based on their contents. 

PRINTALL (or PA) processes multiple conditional updates 
(IF OVERLAY, IF CHANGE, IF MOVE, IF WARP) while printing records.

PRINTREV (or PRR) prints sequential or VSAM data sets in reverse order.

The following example will print the first 10 records:

DD01 PRINT MAXRECOUT=10

The following example will print the first 20 records containing RED in position 12 
replaced with BLU. Note that the input data set is not actually changed; use this 
function to see changes before running an UPDATEREC:

DD01 PRINT OVERLAY=(12,EQ,C'RED',C'BLU'),MAXRECOUT=20

The following example will print the first 100 positions of the first 20 records that 
contain RED in position 12:

DD01 PRINT IF=(12,EQ,C'RED'),MAXRECOUT=20,MOVE=(1,100,1)

The following example will print the first 100 positions of the last 20 records with RED 
in position 12 in reverse order:

DD01 PRINTREV IF=(12,EQ,C'RED'),MAXRECOUT=20,MOVE=(1,100,1)

PRINTCHR, PRINTCHRALL, PRINTCHRMBR 
AND PRINTCHRREV

PRINTCHR (or PC) prints a data set or a portion of a data set in alphanumeric format 
with no additional record information. If you wish to print a ruler at the top of each 
page to help identify the columns used, you can specify the OPTIONS=SHORT 
parameter.

PRINTCHRMBR (or PCM) prints members based on their contents.

PRINTCHRALL (or PCA) processes multiple conditional updates (IF OVERLAY, IF 
CHANGE, IF EXPAND, IF MOVE, IF WARP) while printing records.

PRINTCHRREV (or PCR) prints sequential or VSAM data sets in reverse order.

The following example will list the first 10 records:

DD01 PRINTCHR MAXRECOUT=10

The following example will list the first 20 records containing RED in position 12 
replaced with BLU. Note that the input data set is not actually changed; use this 
function to see changes before running an UPDATEREC:

DD01 PRINTCHR OVERLAY=(12,EQ,C'RED',C'BLU'),MAXRECOUT=20

The following example will list the first 100 positions of the first 20 records that contain 
RED in position 12:

DD01 PRINTCHR IF=(12,EQ,C'RED'),MAXRECOUT=20,MOVE=(1,100,1)

The following example will list the last 10 records in reverse order:

DD01 PRINTCHRREV MAXRECOUT=10

PRINTHEX, PRINTHEXALL, PRINTHEXMBR, 
AND PRINTHEXREV

PRINTHEX (or PH) prints records in a vertical hexadecimal format. The report 
produced also supplies additional useful information such as the record number and 
the record length. A column scale is normally printed with each record; however, if you 
request OPTIONS=SHORT, the column scale will only be produced after each page 
header. 

PRINTHEXMBR (or PHM) prints members based on their contents.

PRINTHEXALL (or PHA) processes multiple conditional updates (IF OVERLAY, IF 
CHANGE, IF EXPAND, IF MOVE, IF WARP) while printing records.

PRINTHEXREV (or PHR) prints sequential or VSAM data sets in reverse order.

To print the first 10 records of a file in vertical hexadecimal:

DD01 PRINTHEX MAXRECOUT=10

The following example will print, in vertical hexadecimal, the first 20 records 
containing RED in position 12 replaced with BLU. Note: the input data set is not 
actually changed; use this function to see changes before using the update functions.

DD01 PRINTHEX OVERLAY=(12,EQ,C'RED',C'BLU'),MAXRECOUT=20

This example will print, in vertical hexadecimal, up to 100 positions of the first 20 
records with RED in position 12:

DD01 PRINTHEX IF=(12,EQ,C'RED'),MAXRECOUT=20,MOVE=(1,100,1)

The following example will print, in vertical hexadecimal, the last 10 records in reverse 
order:

DD01 PRINTHEXREV MAXRECOUT=10

SKIP AND SKIPREV
SKIP (or S) moves the current record pointer forward, and SKIPREV (or SKR) moves 
it backward. This allows large groups of records to be excluded from processing.

When using these functions, remember the following:

• You may not follow a SKIP function with an UPDATEREC function because the 
data set must be closed and reopened while maintaining data set positioning. 

• You can not follow a SKIP function with a reverse function like COPYREV. 
• You can not follow a SKIPREV function with a forward operation like COPYREC.

The following example will skip the first 100 records before a PRINT function:

DD01 SKIP MAXRECIN=100
DD01 PRINT IF=(10,EQ,C'RED')

The following example will skip the first 100 records before an UPDATEREC function:

DD01 UPDATEREC  MAXRECIN=100
DD01 UPDATEREC OVERLAY=(10,EQ,C'RED',C'BLU')

The following example will skip the last 100 records before a PRINTREV function:

DD01 SKIPREV MAXRECIN=100
DD01 PRINTREV IF=(10,EQ,C'RED')

TOTAL 
TOTAL or T reads input records processing all parameter groups for SUM. The 
TOTAL function creates simple reports that can be used to validate data in a data set. 
The results generated from the TOTAL function and the SUM parameter are directed 
to the SYSTOTAL DDNAME if it is present; otherwise, the results are directed to the 
SYSPRINT data set. Comment statements can also be added to the input control 
statements and they will be printed with the TOTAL function results. 

StarBat will produce a two line data set identification message after the comment and 
before the accumulation with the following output identifier: FOLLOWING TOTALS 
DEVELOPED FROM for the first line and the input data set name and volume name on 
the second line. 

The following example will accumulate all packed numbers beginning in column 43:

DD01 TOTAL  SUM=(43,'This is the packed data set total')

The following example will accumulate all character numbers beginning in column 43 
for type M records:

DD01 TOTAL  IF=(22,EQ,C'M'),SUM=(43,6,C,'This is the character data set 
total')

UPDATEREC, UPDATEMBR, AND 
UPDATEALL

UPDATEREC (or UR) updates records in place. This function must be combined with 
any of the parameters used to modify data such as the CHANGE and OVERLAY 
parameters.

Note This will update the data set in place. If you wish to preview your 
changes prior to committing them, use the PRINT functions instead to 
show the changes that will be made and then re-run the job replacing the 
PRINT function with the UPDATEREC function.

UPDATEMBR (or UM) updates members based on their contents.

UPDATEALL (or UA) processes multiple conditional updates (IF OVERLAY, IF 
CHANGE or IF WARP) while processing all records.

The following example will show potential changes before performing an actual 
UPDATEREC:

DD01 PRINT  OVERLAY=(1,0,C'RED',C'BLU')

The following example will change the first occurrence of RED anywhere in a record to 
BLU:

DD01 UPDATEREC OVERLAY=(1,0,C'RED',C'BLU')

The following example will change all occurrences of RED to BLU and all occurrences 
of WHI to ANY:

DD01 UPDATEALL OVERALL=(1,0,C'RED',C'BLU'),
               OVERALL=(1,0,C'WHI',C'ANY')

Parameters may be specified on function statements to limit the data processed and 
to control the function. 

In this section, the following parameters are described in alphabetical order.

Parameter Short Description

ABEND AB Controls EOJ processing when an abnormal 
condition occurs

AND IF Creates a logical AND condition check (used with 
IF)

CHANGE C Changes only the first instance of data in a record

CHANGEALL CA Changes all occurrences of data in a record

COPYOVER CO Controls the replacement of identically named 
output members

EXCLUDEREC XR Controls the # of records to bypass in an 
EXCLUDEREC function

EXPAND EX Expands records at a specified location 

IF AND Selects records to process based on data contents

MAXRECIN MRI Controls the number of records to input

MAXRECOUT MRO Controls the maximum number of records to output

MEMBER M Specifies a member name to process in a PDS

MEMBERS MS Specifies a group of members to process in a PDS 
with a mask

MOVE MV Moves data into the record

NEWMBR NM Gives a new name to an output PDS member

NEWMBRS NMS Names multiple new members of an output PDS 
using a mask

OPTIONS OP Controls StarBat processing options

OR (none) Used with the IF parameter to indicate an OR 
condition

OVERALL OA Replaces all occurrences of data in a record with 
other data

OVERLAY OL Replaces the first instance of data in a record with 
new data

PADCHAR PAD Specifies a padding character for uninitialized parts 
of a record

PRINT P Prints records in alphanumeric format with record 
statistics

PRINTCHR PC Prints records in alphanumeric format

PRINTHEX PH Prints records in vertical hexadecimal format

PRINTLPI PL Specifies the number of lines per inch for print 
output pages

RBA (none) Processes VSAM data beginning at a relative byte 
address

RDW (none) Controls the inclusion of the record descriptor word

SELECT S Processes every nth record

STARTKEY SK Processes VSAM data beginning with a generic key

STOPIF ST Stops processing a function when a record satisfies 
a condition

SUM (none) Accumulates the contents of specified fields

WRITE W Writes a record to one or more output files

Parameter Short Description



37

Serena™ StarTool® FDM 7.4 StarBat Option Getting Started Guide

Parameters are grouped according to type as follows:

• Action Changes data (CHANGE, EXPAND, MOVE, OVERLAY, OVERALL, 
SUM, WARP and WRITE)

• Limit Specifies record count limits (EXCLUDEREC, MAXRECIN, 
MAXRECOUT and SELECT)

• Print Prints records as they are processed (PRINT, PRINTCHR and 
PRINTHEX).

• Selection Selects records based on their contents (AND, IF, and OR)

• Control Defines basic conditions during execution (all other parameters such 
as ABEND and STOPIF)

Several restrictions on parameter and function combinations should be noted:

• The EXCLUDEREC parameter can only be used with the EXCLUDEREC 
function.

• The WRITE parameter can only be used with the MULTICOPY function.

• The PRINTHEX, PRINTCHR and PRINT parameters should not be used with the 
PRINT functions.

• The EXPAND, MOVE and MAXRECOUT parameters can not be used with the 
UPDATE functions.

• The NEWMBR, NEWMBRS and COPYOVER parameters can not be used with 
PRINT functions, SKIP, SKIPREV or UPDATEREC.

Many StarBat parameters process numeric character data and packed decimal data. 
Numeric characters must be in zoned decimal (hexadecimal X'F0' through X'F9') and 
the last byte can be signed positive (X'C0'), negative (X'D0') or unsigned (X'F0'). 
Packed decimal numbers must contain valid numeric numbers and the sign must be 
positive (with either X'0C' or X'0F') or negative (X'0D'). In most cases, you can enter a 
data length of 0 for packed decimal numbers and StarBat will determine the length 
dynamically by scanning for the sign digit in each field as it is processing each record. 

ABEND
The ABEND or AB parameter controls how the end of job processing should be 
handled when an abnormal condition occurs during job execution.

   ABEND=0/1/2

ABEND=0 Issues a return code at normal end of job.

ABEND=1 Issues a U0012 abend dump when an I/O error occurs. This is the 
default.

ABEND=2 Issues a user abend when a non-zero return code is encountered.

In the following example, a user abend of U0008 will occur if the character string 
‘ALIAS-‘ is not found in the input data set and no records are written to the output data 
set.

DD01 MULTICOPY ABEND=2,IF=(1,0,C'ALIAS-').MOVE=(19,0,80),WRITE=FILE1

AND
The AND parameter selects records to be processed by the function being executed. 
The AND parameter and the IF parameter have identical meanings and format. AND 
is normally used after an IF parameter to improve readability.

For additional details, see “IF” on page 42.

CHANGE
The CHANGE or C parameter works like the CHANGE command in ISPF edit. It 
replaces the value in string-1 with the value in string-2. String-1 and string-2 can be 
different lengths and the data will be shifted left or right depending on the size of the 
second string (see NOTE: below). CHANGE will only modify the first occurrence of the 
matching data in the record, as with ISPF. If you wish to change multiple occurrences, 
use the similar CHANGEALL parameter (page 40).

   CHANGE=(start,length/operator,string-1,string-2)

start defines the position of the record where the search is to begin. The first 
byte of the record is position 1.

length specifies how many characters to search. Specify 0 if you want to search 
the remainder of the record. 

operator an alternative to length is to specify a relational operator (such as EQ).

string-1 is a binary, character, hexadecimal or packed string representing data to 
be searched for at this location.

string-2 is a binary, character, hexadecimal or packed string representing data to 
be replaced at this location.

NOTE: When CHANGE replaces a data field by a shorter field, StarBat shifts 
contiguous characters to the left until a blank character is found and 
inserts blanks to adjust for the missing characters. If no blanks are found, 
blanks are inserted at the end of the record. 

For example, 

with CHANGE=(1,5,C'1234',C'X')

  Record 1: 12345 AB1234GH
  Record 2: 123456ABCDEFGH

would become:

  Record 1: X5    AB1234GH
  Record 2: X56ABCDEFGH

When CHANGE replaces a data field by a longer field, StarBat shifts contiguous 
characters to the right compressing multiple blanks to single blanks until all characters 
fit into the record. Note: if no blanks are found using the COPYREC or MULTICOPY 
function for fixed length output records or using the UPDATE functions, truncation will 
occur if the data extends beyond the record boundary. 

For example, 

with CHANGE=(11,3,C'USA',C'DOMESTIC')

  Record 1: AA19980101USA       CA94010-1904
  Record 2: AA19880101USA  CA94010-1904

would become:

  Record 1: AA19980101DOMESTIC  CA94010-1904
  Record 2: AA19880101DOMESTIC CA94010-1904

It is important to remember when using the CHANGE (or OVERLAY) parameters that 
they operate on the input data records. Therefore, you should specify any MOVE 
functions after the CHANGE (or OVERLAY) parameter because MOVE operates on 
the output record. 

CHANGEALL
The CHANGEALL or CA parameter is the same as the CHANGE parameter except 
that all occurrences are modified and the second parameter must be a length value 
(not an operator like EQ). See “CHANGE” on page 39 for additional details.

CHANGEALL=(start,length,string-1,string-2)

start defines the position of the record where the search is to begin. The first 
byte of the record is position 1.

length specifies how many characters to search. Specify 0 if you want to search 
the remainder of the record. 

string-1 is a binary, character, hexadecimal or packed string representing data to 
be searched for at this location.

string-2 is a binary, character, hexadecimal or packed string representing data to 
be replaced at this location.

COPYOVER
The COPYOVER or CO parameter may be used to specify if identically named output 
members are to be replaced when copying from one PDS to another.

   COPYOVER=YES/NO

COPYOVER=YES Default, replaces the output PDS member if it already exists. 

COPYOVER=NO Do not replace the output PDS member if it already exists.

The following example specifies that existing members are not to be replaced: 

DD01 COPYREC COPYOVER=NO

EXCLUDEREC
The EXCLUDEREC or XR parameter is used with the EXCLUDEREC function to 
control the number of records to exclude. 

  EXCLUDEREC=n

n   is a number from 1 to 999999999. It represents the number of records to be 
excluded.

The following example replaces all occurrences of ‘MAIL’ with ‘IMA1’ and omits the 
first record that contains ‘IMA1-M’.

DD01 EXCLUDEREC OVERALL=(1,0,C’MAIL’,C’IMA1’),IF=(1,EQ,C’IMA1-M’),EXCLUDEREC=1

EXPAND
The EXPAND or EX parameter is used to expand date and data fields. EXPAND is 
similar to MOVE but the original record is preserved to the extent possible. Character 
type is normally used for date expansions of any format because after a date is 
padded on the right with blanks; StarBat can still read it using its input picture and 
rewrite it with a new picture.

Column numbers referred to by EXPAND are the original column numbers in the data; 
similarly, references to column numbers by other StarBat parameters should also 
reference the original column numbers because StarBat will adjust these column 
numbers dynamically. 

   EXPAND=(column,type,size,newsize)

column is the location where the data item begins. Any valid actual or relative 
location may be used. 

type is B for binary, P for packed, C for character or N for character numeric. 
Character type pads the original item on the right with blanks; other types 
of field expansion add numeric padding to the left.

size is the number of bytes used by the original data item. 

newsize is the number of bytes to be used by this data item. Newsize can be 
smaller or larger than size. 

The following example will expand a packed data field and accumulate the next 
character field: 

DD01 COPYREC  EXPAND=(21,P,4,8),SUM=(25,5,C,'The next field')

The following example will expand a date field and convert a character date format 
from YYDDD to CCYYDDD: 

DD01 COPYREC  EXPAND=(21,C,5,7),WARP=(21,C,YYDDD,OUTPIC=CCYYDDD)

The following example will expand a date field and convert a packed date format from 
YYDDD to CCYYDDD: 

DD01 COPYREC  EXPAND=(21,C,3,4),WARP=(21,P,YYDDD,OUTPIC=CCYYDDD)

The following example will expand a binary field and a numeric character field and 
accumulate the fields: 

DD01 COPYREC  
EXPAND=(12,B,2,4),EXPAND=(21,N,4,9),SUM=(12,4,B),SUM=(21,9,C)

IF
The IF parameter selects records to be processed by the function being executed. 
The IF parameter and the AND parameter have identical meanings and format. AND 
is normally used after an IF parameter to improve readability. 

There are two IF syntax forms to test for data contents or valid numerics:

   IF=(start,length/operator,string,...)     /* data content test   */
   IF=(start,length,[duplicate]type,...)     /* valid numeric test  */

start defines the position of the record where the search is to begin. The 
first byte of the record is position 1.

length specifies how many characters to search. Specify 0 if you want to 
search the remainder of the record. 

operator an alternative to length is to specify a relational operator (such as 
EQ).

string is a binary, character, hexadecimal or packed string representing 
data to be searched for at this location.

[duplicate]type is an optional duplication number followed by EQN and EQP to 
check for valid numeric character or packed decimal or NEN and 
NEP to check for invalid numeric character or packed decimal.

Note You may enter multiple parameter sets in an IF parameter separated by 
commas to obtain multiple logical OR tests.

In the following example, all records without a hex ‘F0F1’ in column 1 are written to file 
FILE1.

DD01 MULTICOPY IF=(1,NE,X’F0F1’),WRITE=FILE1

The following example copies all records, but in those records having an ‘ME’ in 
position 1 AND a ‘QE’ in column 3, position 10 will be replaced with a ‘P’.

DD01 COPYALL IF=(1,EQ,T’ME’),IF=(3,EQ,C’QE’),OVERLAY=(10,C’P’) 

The example above can also be coded using the AND parameter.

DD01 COPYALL IF=(1,EQ,T’ME’),AND=(3,EQ,C’QE’),OVERLAY=(10,C’P’) 

The following example directly updates the input data set. If the data at position 60 for 
a length of 5 is not numeric, position 60 will be replaced with five zeros.

DD01 UPDATEREC IF=(60,5,NEN),OVERLAY=(60,C’00000’)

In the following example, all records with invalid packed numbers will be copied to the 
output data set: 

DD01 COPYREC IF=(44,0,NEP),PRINTHEX=4

This example will copy input records with a 1 in position 10 and a 2 in position 20, or 
an A in position 3 and a B in position 6. Since it is a COPYREC, only position 30 would 
be warped; COPYSOME or COPYALL would warp both positions. 

DD01 COPYREC IF=(10,EQ,C’1’),AND=(20,EQ,C’2’),WARP=(30,C,CCYYMMDD,ADD=5Y),PRINT=3,
              IF=(3,EQ,C’A’),AND=(6,EQ,C’B’),WARP=(40,C,CCYYMMDD,ADD=10D),MAXRECOUT=30 

The following example will print, in vertical hex, all records that contain odd EBCDIC 
numbers in column 14: 

DD01 PRINTHEX IF=(14,AO,X'F1')

The following example will print, in vertical hex, all records that contain even EBCDIC 
numbers in column 14: 

DD01 PRINTHEX IF=(14,GE,X'F0'),IF=(14,NO,X'01')

The following example will print, in vertical hex, all records that contain an EBCDIC 0, 
4 or 8 in column 14: 

DD01 PRINTHEX IF=(14,AO,X'F0'),IF=(14,AZ,X'03')

MAXRECIN
The MAXRECIN or MRI parameter specifies the number of input records to be read 
before terminating function processing.

   MAXRECIN=n

n   is the maximum records to be input. Any number from 0 to 999999999 can be 
used where 0 is all records.

The following example will copy the first 100 records from the input data set to the 
output data set: 

DD01 COPYREC MAXRECIN=100

MAXRECOUT
The MAXRECOUT or MRO parameter controls the number or records to be printed or 
written before processing stops. MAXRECOUT can be used to limit the number of 
output records or extend the number of printed records past the default of 250. Note 
that the MAXRECOUT parameter is ignored for the update functions.

   MAXRECOUT=n

n   is the maximum records to output. Any number from 0 through 999999999 may be 
used where 0 is all records.

The following example will print, in vertical hex, the first 30 records that contain the 
string TEST RECORD in column 23: 

DD01 PRINTHEX IF=(23,EQ,C'TEST RECORD'),MAXRECOUT=30 

MEMBER
The MEMBER or M parameter specifies a member name to process in a PDS. If 
neither MEMBER nor MEMBERS is specified for a PDS, all members of the data set 
will be processed. 

   MEMBER=memname/(member1,member2,…)

memname specifies the member to be processed. To specify multiple members, 
use parentheses with MEMBER; use the MEMBERS parameter; or use 
multiple control statements with MEMBER and the same DDxx name. 

The following example will copy a single member.

DD01 COPYREC MEMBER=MEMBERA

The following example will print three different members in vertical hex.

DD01 PRINTHEX MEMBER=MEMBERA
DD01 PRINTHEX MEMBER=MEMBERB
DD01 PRINTHEX MEMBER=MEMBERC

The following example will print three different members in vertical hex.

DD01 PRINTHEX MEMBER=(MEMBERA,MEMBERB,MEMBERC)

MEMBERS
The MEMBERS or MS parameter specifies a group of members to process in a PDS 
with a mask. If neither MEMBER nor MEMBERS is specified for a PDS, all members 
of the data set will be processed. 

   MEMBERS=ALL/maskname

ALL specifies that all members are to be selected. 

maskname specifies the members to be processed using a mask of up to eight 
characters which is matched with member names from the data set. 
Several mask examples are shown below; for more information, see 
“Member Name Forms” in the StarTool FDM Reference Guide. 

The following example will copy all members whose names begin with STAR.

DD01 COPYREC MEMBERS=STAR

Dash: Copy all members whose names begin with STAR and contain 98 in positions 
seven and eight.

DD01 COPYREC MEMBERS=STAR--98

Combination: Copy all members whose names begin with STAR and contain 98 
elsewhere.

DD01 COPYREC MEMBERS=STAR*98

Range: Copy all members in the alphabetic range whose names begin with ABC 
through DEF999999.

DD01 COPYREC MEMBERS=ABC:DEF

Pattern: Copy all members whose names contain STAR and 98 anywhere.

DD01 COPYREC MEMBERS=STAR/98

MOVE
The MOVE or MV parameter builds an output record by moving data to it. Data from 
control statements or from input records can be used depending on which form of the 
MOVE parameter is used. Note: MOVE can not be used with the update functions.

There are two MOVE syntax forms that move data from a control statement or from an 
input record:

MOVE=(toposition,data)                 Comment: data from control statement
MOVE=(toposition,length,from-loc)    Comment: data from input record

toposition defines the starting position in the output record to which the data is 
to be moved. Absolute and relative positions can be used. ‘n’ means 
at position ‘n’, ‘+n’ means ‘n’ bytes further down the record, ‘-n’ 
means ‘n’ bytes previous in the record and ‘+0’ means in the next 
available position.

data is a binary, character, hexadecimal or packed string representing 
data to be moved to the output record.

length defines the number of bytes to be moved. The number can be in the 
range 1 to 255. A value of 0 indicates that the remainder of the input 
record is to be moved.

fromposition defines the starting position in the input record from which the data is 
to be moved. Absolute and relative positions can be used as for 
toposition; ‘+0’ references the input relative location or the last scan 
position.

The output record area is initially set to X’00’ by default but this can be overridden by 
the PADCHAR parameter. A data value set in the output record remains set until new 
data is placed there; the output record does not need to be reinitialized.

The following example will move REDRED to the first six positions of an output record.

DD01 COPYREC MOVE=(1,2C'RED')

The following example will print the first 100 characters of each input record.

DD01 PRINT MOVE=(+0,100,+0)

The following example will substitute RED for the first three positions of each record.

DD01 COPYREC MOVE=(1,C'RED'),MOVE=(+0,0,4)

NEWMBR
NEWMBR or NM names a new member of an output PDS.

   NEWMBR=memname

memname specifies a new name to be assigned to the copied member.

The following example will copy MEMBERA to an output data set and rename it to 
MEMBERCC.

DD01 COPYREC MEMBER=MEMBERA,NEWMBR=MEMBERCC

Note that an existing member will be replaced unless the COPYOVER=NO parameter 
is specified.

DD01 COPYREC MEMBER=MEMBERA,NEWMBR=MEMBERCC,COPYOVER=NO

NEWMBRS
NEWMBRS or NMS assigns new names to a group of output PDS members.using a 
mask.

   NEWMBRS=maskname

maskname The maskname can be from 1 to 8 characters long. Characters in the 
member name are replaced by characters from the maskname unless 
the character in the maskname contains a minus sign; the character in 
that position will be retained. Any blanks in the resulting member names 
are discarded.

The following example selects all members prefixed with TST and renames them to 
SYS..A. For example, member TST3333 will get renamed to SYS33A3. Members that 
contain the same name in the output data set will not be replaced. 

DD01 COPYMBR MEMBERS=TST,NEWMBRS=SYS--A,COPYOVER=NO  

This example will copy those members prefixed with TST and containing ‘ALPHA’ in 
column 1 in the data, and renames the selected members to PARM…

DD01 COPYMBR MEMBERS=TST,IF=(1,EQ,C’ALPHA’),NEWMBRS=PARM   

OPTIONS
OPTIONS or OP controls StarBat processing options.

   OPTIONS=LONG/SHORT/JCL/MULTI/NOMULTI

OPTIONS=LONG default, a column scale is to be printed for PRINT and 
PRINTHEX after each data record.

OPTIONS=SHORT a column scale is only to be printed at the top of each PRINT, 
PRINTCHR and PRINTHEX page.

OPTIONS=JCL logical JCL processing is desired for DD, EXEC, JOB, PROC 
and SET JCL statements. 

OPTIONS=MULTI after an end of file condition, a function with the same 
DDNAME is to reread the data set.

OPTIONS=NOMULTI default; after an end of file condition, another read will note 
another end of data set condition.

The following example will print a column scale only on each page.

DD01 COPYREC OPTIONS=SHORT,MEMBER=MEMBERA,PRINTHEX=500

The following example will utilize logical JCL processing for this control statement 
group.

DD01 COPYREC 
OPTIONS=JCL,IF=(1,0,C’VOL=SER=’),CHANGE=(1,0,C’UNIT=TAPE3’,C’UNIT=TAPE77’)

The following example will cause StarBat to reread the data set if the end of the data 
set was encountered.

DD01 COPYREC OPTIONS=MULTI,PRINTHEX=500

OR
Provides for those situations where a number of conditions need to be grouped 
together with the OR operator. There are two OR syntax forms to test for data 
contents or valid numerics:

   OR=(start,length/operator,string,...)     /* data content test   */
   OR=(start,length,[duplicate]type,...)     /* valid numeric test  */

start defines the position of the record where the search is to begin. The 
first byte of the record is position 1.

length specifies how many characters to search. Specify 0 if you want to 
search the remainder of the record. 

operator an alternative to length is to specify a relational operator (such as 
EQ).

string is a binary, character, hexadecimal or packed string representing 
data to be searched for at this location.

[duplicate]type is an optional duplication number followed by EQN and EQP to 
check for valid numeric character or packed decimal or NEN and 
NEP to check for invalid numeric character or packed decimal.

Note You may enter multiple parameter sets separated by commas in an OR 
parameter to obtain multiple logical OR tests.

In this example, if position 40 contains ‘COPYRIGHT’ or ‘247B’ anywhere from 
column 2 through the end of the record, copy the record to the file identified in the JCL 
by FILEA.

DD01 MULTICOPY IF=(40,EQ,C’COPYRIGHT’),OR=(2,0,C’247B’),WRITE=FILEA

This example copies all records that contain a packed ‘117400477’ or ‘516988271’ in 
position 11.

DD01 COPYREC IF=(11,EQ,{‘117400477’),OR=(11,EQ,P’516988271’),MAXRECOUT=0 

This example copies all records, but if column 1 contains a hex ‘22722D’ OR a hex 
‘28888C’, replace column 1 with a hex ‘99999C’.

DD01 COPYALL IF=(1,EQ,X’22722D’,1,EQ’X’28888C’),OVERLAY=(1,X’99999C’)

The example above is equivalent to

DD01 COPYALL IF=(1,EQ,X’22722D’),OR=(1,EQ,X’28888C’),OVERLAY=(1,X’99999C’)

The following example will copy all type 1 records or records with invalid packed 
numbers to the output data set: 

DD01 COPYREC IF=(12,EQ,C'1'),OR=(44,0,NEP),PRINTHEX=4

This example will copy input records with a 1 in position 10 and a 2 in position 20 or an 
A in position 3 and a B in position 6. Since it is a COPYREC, only position 30 would be 
warped; COPYSOME or COPYALL would warp both positions. 

DD01 COPYREC IF=(10,EQ,C’1’),AND=(20,EQ,C’2’),WARP=(30,C,CCYYMMDD,ADD=5Y),PRINT=3,
            IF=(3,EQ,C’A’),AND=(6,EQ,C’B’),WARP=(40,C,CCYYMMDD,ADD=10D),MAXRECOUT=30 

OVERLAY
The OVERLAY or OL parameter replaces data at a particular location with new data in 
a variety of ways. Data in one location can be overlaid with literal data unconditionally, 
according to some conditional data content in a location, or according to some 
conditional data content in another separate location. 

There are three OVERLAY syntax forms: 

OVERLAY=(position,newstring)                                    Replace by location
OVERLAY=(position,length/operator,string-1,string-2)            Replace by condition
OVERLAY=(position,length/operator,string-1,toposition,string-2) Replace at alternate loc

position defines the starting position in the output record to which the data is to 
be moved. Absolute and relative positions can be used. ‘n’ means at 
position ‘n’, ‘+n’ means ‘n’ bytes further down the record, ‘-n’ means ‘n’ 
bytes previous in the record and ‘+0’ means in the next available 
position. 

newstring is a binary, character, hexadecimal or packed string representing data 
to be placed in the specified location.

length specifies how many characters to search. Specify 0 if you want to 
search the remainder of the record. 

operator an alternative to length is to specify a relational operator (such as EQ).

string-1 is a binary, character, hexadecimal or packed string representing data 
to be searched for at this location.

toposition defines the starting position in the output record to which the data is to 
be moved. Absolute and relative positions can be used. ‘n’ means at 
position ‘n’, ‘+n’ means ‘n’ bytes further down the record, ‘-n’ means ‘n’ 
bytes previous in the record and ‘+0’ means in the next available 
position.

string-2 is a binary, character, hexadecimal or packed string representing data 
to be replaced at this location.

  !!Caution!! OVERLAY replaces existing data with the new data at 
the specified location without shifting data.

Note that bit operations can be indicated by specifying a modifier as the first character 
of string-2; O indicates an OR, M indicates NOT-AND or binary minus, A indicates 
AND and E indicates an exclusive OR.

The following example will copy all records and place REDRED in the first 6 positions 
of each record.

DD01 COPYREC OVERLAY=(1,2C'RED')

The following example will replace the first A found in column 10 through 19 with a B.

DD01 COPYREC OVERLAY=(10,10,C'A',C'B')

The following example will find RED beginning in position 6 and overlay position 10 
with an asterisk.

DD01 COPYREC OVERLAY=(6,EQ,C'RED',10,C'*')

The following example will clear the zone digits where ABCD is found with an AND bit 
operation.

DD01 COPYREC OVERLAY=(23,20,C'ABCD',AX'0F0F0F0F')

OVERALL
The OVERALL or OA parameter replaces data at a particular location in the same way 
as OVERLAY does; however, OVERALL replaces all occurrences of the string and it 
only supports a length field. For more information, see OVERLAY.

There are two OVERALL syntax forms: 

OVERLAY=(position,length,string-1,string-2)              Note: Replace by condition
OVERLAY=(position,length,string-1,toposition,string-2)   Note: Replace at alternate loc

position defines the starting position in the output record to which the data is to be 
moved. Absolute and relative positions can be used. ‘n’ means at position 
‘n’, ‘+n’ means ‘n’ bytes further down the record, ‘-n’ means ‘n’ bytes 
previous in the record and ‘+0’ means in the next available position. 

length specifies how many characters to search. Specify 0 if you want to search 
the remainder of the record. 

string-1 is a binary, character, hexadecimal or packed string representing data to 
be searched for at this location.

toposition defines the starting position in the output record to which the data is to be 
moved. Absolute and relative positions can be used. ‘n’ means at position 
‘n’, ‘+n’ means ‘n’ bytes further down the record, ‘-n’ means ‘n’ bytes 
previous in the record and ‘+0’ means in the next available position.

string-2 is a binary, character, hexadecimal or packed string representing data to 
be replaced at this location.

The following example will replace all RED's found in columns 10 through 19 with 
BLU's.

DD01 COPYREC OVERALL=(10,10,C'RED',C'BLU')

PADCHAR
The PADCHAR or PAD parameter is used to initialize an output area with the 
specified character. The default is binary zeros (X’00’).

PADCHAR=C’c’/X’nn’

C’c’ c is any single character value.

X’nn’ nn is any valid hexadecimal value.

If a record has 10 blanks beginning in position 11, move the indicated data and pad 
the end of the records with periods.

DD01 COPYREC IF=(11,EQ,10C’ ‘),MOVE=(1,5,28),PADCHAR=C’.’ 

PRINT
The PRINT or P parameter prints a specified number of records in a simple character 
format. Note: the PRINT parameter is not intended for use with the print functions; it 
should be used with any of the other functions that process the data set.

   PRINT=n

n Any number from 0 to 999999999. Specify 0 to print all records.

The following example will print the first 10 records that contain the string RED 
anywhere.

DD01 COPYREC IF=(1,0,C'RED'),PRINT=10

PRINTCHR
The PRINTCHR or PC parameter prints a specified number of records in an 
alphanumeric format. 

Note PRINTCHR is not intended for use with the print functions; it should be 
used with any of the other functions that process the data set. 

   PRINTCHR=n

n Any number from 0 to 999999999 can be used. Specify 0 to print all records.

The following example will print the first 10 records that contain the string RED 
anywhere.

DD01 COPYREC IF=(1,0,C'RED'),PRINTCHR=10

PRINTHEX
The PRINTHEX or PH parameter prints a specified number of records in vertical 
hexadecimal format. Note: PRINTHEX is not intended for use with the print functions; 
it should be used with any of the other functions that process the data set. 

   PRINTHEX=n

n Any number from 0 to 999999999. Specify 0 to print all records.

The following example will print in vertical hex the first 10 records that contain the 
string RED anywhere.

DD01 COPYREC IF=(1,0,C'RED'),PRINTHEX=10

PRINTLPI
The PRINTLPI or PL parameter controls the number of lines written to a page for the 
SYSLIST output data set.

   PRINTLPI=6/8

PRINTLPI=6 default, specifies 6 lines to an inch and allows up to 58 lines in a 
SYSLIST output. 

PRINTLPI=8 specifies 8 lines to an inch and allows up to 78 lines in a SYSLIST 
output.

The following example will print the first 10 records containing the string RED with as 
many as 78 lines on a page.

DD01 PRINTHEX IF=(1,0,C'RED'),PRINTCHR=10,PRINTLPI=8

RBA
The RBA parameter is used for VSAM data sets. For ESDS and KSDS, this defines 
the Relative Byte Address. For RRDS, this defines the Relative Record Number.

   RBA=nn/X’nn’

nn nn is any decimal value.

X’nn’ nn is any valid hexadecimal value.

The example below lists records beginning at relative byte address 800. The relative 
byte address of 800 can also be specified in hex as X’320’.

DD01 PRINTCHR RBA=800  

RDW
The RDW (Record Descriptor Word) parameter specifies how to address the 
RECFM=V RDW word.

   RDW=0/1/2/3

RDW=0 default, the RDW word is to be included for record processing and 
output displays.

RDW=1 the RDW word is to be included for record processing but not for output 
displays.

RDW=2 the RDW word is not to be included for record processing but is to be 
shown for output displays.

RDW=3 the RDW word is not to be included for record processing and not to be 
shown for output displays.

The following example lists the first 10 records that contain RED in the first data 
position and the RDW word is to be included for record processing and output: 

DD01 PRINTCHR RDW=0,MAXRECOUT=10,IF=(5,EQ,C'RED')

The following example is similar, looking for records with RED in the first data position. 
Note that the RDW word is not to be displayed in the output: 

DD01 PRINTCHR RDW=3,MAXRECOUT=10,IF=(1,EQ,C'RED')

SELECT
The SELECT or S parameter may be used to select every nth occurrence of a record 
for processing.

   SELECT=n

n is the number of records in an interval. Any number from 1 to 999999999 may 
be used.

The following example copies all records and tests every third record for a ‘PBCE’ in 
column 3. If a match is found, ‘PBCE’ is replaced by four zeroes.

DD01 COPYALL SELECT=3,OVERLAY=(3,EQ,C’PBCE’,C’0000’)

The following example only processes records containing ‘Z145’ in column 3. Every 
third record satisfying the conditional statement will be copied and have ‘Z145’ 
replaced with ‘ZZZZ’.

DD01 COPYREC IF=(3,EQ,C’Z145’),SELECT=3,OVERLAY=(3,C’ZZZZ’)

In this example, every third record is selected, and if the selected record contains a 
‘PBCE’ in column 3, copy the record and replace ‘PBCE’ with ‘ZZZZ’.

DD01 COPYREC SELECT=3,IF=(3,EQ,C’PBCE’),OVERLAY=(3,C’ZZZZ’)      

STARTKEY
The STARTKEY or SK parameter can be specified to initiate VSAM KSDS processing 
at a specific or generic key.

   STARTKEY=C'cc'/X'hh'

STARTKEY=C'cc' is any character string that specifies the beginning KSDS key 
value.

STARTKEY=X'hh' is any hexadecimal string that specifies the beginning KSDS key 
value.

The following example will copy VSAM records after positioning to generic key RED: 

DD01 COPYREC STARTKEY=C'RED'

STOPIF
Use the STOPIF or ST parameter if you want to terminate processing before the end 
of data. STOPIF will end execution of the current process if the condition associated 
with the parameter is satisfied.

   STOPIF=(start,length/operator,string,...)

start defines the position of the record where the search is to begin. The first 
byte of the record is position 1.

length specifies how many characters to search. Specify 0 if you want to 
search the remainder of the record. 

operator an alternative to length is to specify a relational operator (such as EQ).

string is a binary, character, hexadecimal or packed string representing data 
to be searched for at this location.

Note You may code multiple conditions within one STOPIF parameter; these 
are evaluated using logical OR tests. Multiple contiguous STOPIF 
parameters result in logical AND tests for STOPIF conditions.

The following example will copy all records until ‘RECORDS READ’ is found in 
position 8. The record containing ‘RECORDS READ’ is not copied.

DD01 COPYREC STOPIF=(8,EQ,C’RECORDS READ’)

In this example, the first DD01 control statement will copy records until VOL001 is 
found in column 1. The second DD01 control statement will begin with the VOL001 
record and stop copying records with VOL300.

DD01 COPYREC STOPIF=(1,EQ,C’VOL001’)
DD01 COPYREC STOPIF=(1,EQ,C’VOL300’)

The following example will skip to the record containing a hex ‘033198’ in column 1 or 
‘IMAGE’ in column 12. This record and all records after it are copied.

DD01 SKIP STOPIF=(1,EQ,X’033198’,12,EQ,C’IMAGE’)
DD01 COPYREC

SUM
The SUM parameter is used to add up any fields containing character, packed, or 
binary numbers. If there is a SYSTOTAL data set, then the totals will go to that data 
set. Otherwise, they will go to the SYSPRINT data set

SUM=(position[,'description])                   /* for packed decimal data      */
SUM=(position,length,datatype[,'description'])  /* for character or binary data */

position  The beginning position of the data. This can be an actual or a relative 
start position.

length The length of the data to be totaled. For character data, specify a 
length of 1 to 15; for binary data, specify a length of 1 to 4; for packed 
data, its length is calculated by StarBat.

datatype The type of data. For character data, specify C; for binary data, 
specify B; for packed data, you do not need to specify the type.

description This is optional. A maximum of 25 characters enclosed in single 
quotes can be entered. If a description is present, it will be displayed 
as report headers; otherwise, the control statement will be used as a 
report header. 

The following example selects records that contain a packed decimal ‘9802’ in 
position 5, then adds the packed field at position 12. These selected records are 
printed in vertical format in hex. 

DD01 PRINTHEX IF=(5,EQ,P’9802’),SUM=(12,’HIST TOTAL’)

The example below begins processing a VSAM KSDS data set with the record at 
location x’00433C’ and then adds the data at position 73. Processing stops if the key 
is greater than x’00433C’.

DD01 TOTAL KEY=X’00433C’,
          STOPIF=(1,GT,X’00433C’), 
          SUM=(073,8,C,’TRAVEL EXPENSE REPORT’)     

WRITE
The WRITE or W parameter is used only by the MULTICOPY function to control when 
output is written to the named DD statements. The WRITE parameter can direct 
output to any number of output DD statements; if the same DDNAME is used more 
than once, multiple copies of the output record will be written.

There are several restrictions on the data that can be output with a MULTICOPY 
function:

Output to load members is supported; however JCL indicating PDS(MEMBER) 
notation is not supported. Instead, use only the PDS name in the JCL and indicate the 
member name with the MEMBER, MEMBERS, NEWMBR or NEWMBRS parameters. 

Overlay, scatter-loaded and note-listed load members are not supported.

   WRITE=ddname1/(ddname1,ddname2,...)

ddnamex the DDNAME in the JCL containing the name of the output data set.

The following example shows how to create two identical output data sets in //OUT1 
and //OUT2:

DD01 MULTICOPY WRITE=(OUT1,OUT2)

The following example will copy a file and repeat any record with a C'5' and change it 
to a C'6':

DD01 MULTICOPY WRITE=OUT1,IF=(20,EQ,C'5'),MOVE=(1,0,1),
            MOVE=(20,C'6'),WRITE=OUT1

The following example will add a STEPLIB JCL statement after any EXEC statement:

DD01 MULTICOPY WRITE=NEWJCL,IF=(1,20,C' EXEC '),MOVE=(1,80C' '),
            MOVE=(1,C'//STEPLIB  DD  DISP=SHR,'),
            MOVE=(+0,C'DSN=MYHILEV.MYMIDLEV.MYLOWLEV'),WRITE=NEWJCL

LOGICAL AND CONDITIONS

Consecutive IF parameters represent a logical AND condition. In other words, 
IF=(1,EQ,C’A’),AND=(10,EQ,C’B’) has the same effect as 
IF=(1,EQ,C’A’),IF=(10,EQ,C’B’). 

This example will copy all records containing TEST RECORD in position 23 and RED 
anywhere after position 10: 

DD01 COPYREC IF=(23,EQ,C'TEST RECORD'),IF=(10,0,C'RED')

This example is equivalent because AND and IF are identical in meaning:

DD01 COPYREC IF=(23,EQ,C'TEST RECORD'),AND=(10,0,C'RED')

LOGICAL OR CONDITIONS
Multiple data comparisons can be coded within an IF, AND, or OR parameter. A 
logical OR condition is assumed between multiple data comparisons; logical OR 
conditions can be coded in multiple different formats as shown below.

The following example will copy all records containing TEST RECORD in position 23 
or RED anywhere after position 10:

DD01 COPYREC IF=(23,EQ,C'TEST RECORD',10,0,C'RED')

The following example is identical in effect to the previous example:

DD01 COPYREC IF=(23,EQ,C'TEST RECORD'),OR=(10,0,C'RED')

The following example will copy up to 10 records containing TEST RECORD in 
position 23 and RED anywhere after position 10 or the first 20 records containing A in 
position 3 and B in position 6:

DD01 COPYREC IF=(23,EQ,C'TEST RECORD'),AND=(10,10,C'RED'),MAXRECOUT=10,
             IF=(3,EQ,C'A'),AND=(6,EQ,C'B'),MAXRECOUT=20

PROCESSING MULTIPLE SELECTION PARAMETERS
The IF, AND, and OR parameters are called selection parameters because they are 
used to select records to be processed. Limit parameters like MAXRECIN and 
MAXRECOUT can follow the selection parameters to apply processing record limits 
and action parameters like CHANGE or WARP may be specified to modify or inspect 
data fields. In addition, print parameters like PRINTHEX may be specified to 
selectively print records and control parameters like ABEND or STARTKEY can be 
used to select basic execution choices. 

Complex selection criteria can be specified using multiple selection parameters. 
Multiple selection parameters are handled in groups where a group of selection 
parameters starts with the first selection parameter and ends after any associated 
limit, action and print parameters that follow the selection parameters. 

For example; 

DD01 COPYREC IF=(10,EQ,C’1’),AND=(20,EQ,C’2’),WARP=(30,C,CCYYMMDD,ADD=5Y),PRINT=3,
            IF=(3,EQ,C’A’),AND=(6,EQ,C’B’),WARP=(40,C,CCYYMMDD,ADD=10D),MAXRECOUT=30 

In this example, there are two groups of selection parameters. The first group consists 
of the first IF parameter and the 
first AND parameter and is followed by the first WARP action parameter and the 
PRINT parameter. The second group consists of the second IF parameter and the 
second AND parameter and is followed by the second WARP action parameter and 
the MAXRECOUT limit parameter. 

A logical OR condition is assumed between groups of selection parameters that are 
followed by limit, action or print parameters. This means that if an input record 
contains data that matches any one of the groups of selection parameters then that 
input record is selected for processing. In the example above, records that contain a 1 
in column 10 and a 2 in column 20 are copied as well as records that contain an A in 
column 3 and a B in column 6. 

If an input record contains data that matches with a group of selection parameters, 
then the limit, action and print parameters that follow that group are processed and 
StarBat skips to the next processing group. Whether StarBat actually processes those 
parameters depends on the name of the current function. If the function has an ALL or 
SOME suffix as a part of its name (for example, COPYSOME or COPYALL versus 
COPYREC), StarBat will process all selection groups. Otherwise, any subsequent 
selection groups will be bypassed after the first selection group is satisfied. 

In the example above, since COPYREC is the function, if an input record had a 1 in 
column 10 and a 2 in column 20 and an A in column 3 and a B in column 6, then the 
record would be copied, but only column 30 would be warped. If COPYSOME or 
COPYALL were used then both column 30 and column 40 would be warped. 



61

Serena™ StarTool® FDM 7.4 StarBat Option Getting Started Guide

SELECTING MEMBERS BY CONTENT
Members can be selected by their contents. Use a MBR form of a function name (for 
example, COPYMBR, PRINTMBR, PRINTHEXMBR, PRINTCHRMBR or 
UPDATEMBR) and members that meet the selection criteria will be processed by the 
function. 

The following example will copy all members that have TEST RECORD in position 23 
in one or more records:

DD01 COPYMBR IF=(23,EQ,C'TEST RECORD')

SELECTING MEMBERS BY NAME
Members of a PDS can be selected by their name or by a mask that represents their 
name. The MEMBER and MEMBERS parameter is used to achieve this. 

The MEMBER parameter can only be entered once on a control statement. To specify 
multiple members, use parentheses with MEMBER; use the MEMBERS parameter; or 
use multiple control statements with MEMBER and the same DDxx name.

The following example will copy a single member.

DD01 COPYREC MEMBER=MEMBERA

The following example will print two different members.

DD01 PRINTHEX MEMBER=MEMBERA
DD01 PRINTHEX MEMBER=MEMBERC

The following example will print two different members.

DD01 PRINTHEX MEMBER=(MEMBERA,MEMBERB)

The following example will copy all members whose names begin with STAR.

DD01 COPYREC MEMBERS=STAR

The following example will print all members whose names begin with STAR and 
contain Z in position eight.

DD01 PRINTHEX MEMBERS=STAR---Z

OPTIONS=JCL PROCESSING
StarBat supports OPTIONS=JCL. This means that StarBat can determine JCL control 
statements and continuations; and each statement can be processed logically. 
StarBat recognizes DD, EXEC, JCLLIB, JOB, OUTPUT, PROC, SET JCL statements 
and their logical continuations.

If a line in the member is not one of these statements or its continuation, StarBat will 
use normal change and update processing rules for that single line. Following lines 
can then be searched or changed in either JCL or normal mode. JCL statements must 
begin with a // in column 1 and comment statements are considered a part of the 
concatenation.

Thus, the following would be considered a single statement:

  //STEPABC  EXEC  PGM=StarBat,PARM='BATCH',
  //* THIS IS A COMMENT IN THE MIDDLE OF A CONTINUED STATEMENT
  //  TIME=4,REGION=4096K

To search a JCL structure, use the IF or OR statements. You can also search and 
update JCL statements with the CHANGE, CHANGEALL, OVERLAY and OVERALL 
statements. Be sure to use a length of zero for the search length of these parameters 
if you wish to search the entire JCL statement.

The OVERLAY and OVERALL parameters update data but do not shift data columns 
so they can be used without problems by any of the COPY operations, MULTICOPY 
or the UPDATE operations.

CHANGE and CHANGEALL can cause data shifting if the search and replacement 
strings differ in length. If the string is shortened, StarBat follows normal JCL rules and 
is able to update any of the COPY, MULTICOPY or UPDATE operations. If the string 
is expanded, StarBat adjusts the JCL statement to fit the statement in the new string 
using normal JCL rules.

JCL expansion is performed using the following logic to minimize changes in JCL 
statements:

1. If the string will fit, the JCL text is expanded to the right on the changed JCL line.

2. If it does not fit and the string will fit, the JCL text is added after shifting the 
changed JCL line left and right.

3. If both of the above expansions fail, the JCL line to be changed is split at the 
previous comma and continued on another line and the added line is expanded as 
needed. The new JCL line is indented to match the split JCL line.

4. If the above expansion fails because there is not a previous parameter, StarBat 
scans for the next comma to the right and splits the JCL statement after that 
parameter. Again, the new JCL line is indented to match the split JCL line.

StarBat can not split a JCL line for UPDATE operations. In this case, an expansion 
error will be indicated and the return code for the job will be set to eight.

Restrictions:

• A JCL statement may contain up to 50 JCL lines.

• OPTIONS=JCL and MOVE from input to output is not supported.

• The JCL is assumed to be syntactically correct; however, StarBat does not issue 
syntax messages for incorrect JCL.

• To be considered JCL, the statements must reside in a PDS member and the data 
set must have a fixed record format with 80 character records.

• Only data between columns 1 through 71 are considered for determining control 
statement boundaries and continuations.

The examples in this section illustrate various uses of StarBat functions and 
parameters.

Example 1: This example will copy all records and test every third record for a ‘PBCE’ 
in column 3. If a match is found, ‘PBCE’ will be replaced with 4 zeroes.

DD01 COPYALL SELECT=3,OVERYLAY=(3,EQ,C’PBCE’,C’0000’)

Example 2: This example will replace ‘VOL’ with 3 blanks if ‘DUMP’ is found 
anywhere in columns 1 through 6 AND if ’VOL’ is found 6 relative positions from the 
location of the character ‘D’ in ‘DUMP’. The record will then be copied to the output 
data set.

DD01 COPYREC IF=(1,6,C’DUMP’),IF=(+6,EQ,C’VOL’),OVERLAY=(+0,3C’ ’) 

Example 3: This example will build new records by moving data with MOVE 
parameters. The first 5 records are printed in hexadecimal format.

DD01 COPYREC MOVE=(1,10,5),MOVE=(11,10,20),PRINTHEX=5

Example 4: The first DD01 control statement will copy records until ‘VOL001’ is found. 
The second DD01 control statement begins with the ‘VOL001’ record and stops 
copying when 'VOL300’ is found.

DD01 COPYREC STOPIF=(1,EQ,C’VOL001’)
DD01 COPYREC STOPIF=(1,EQ,C’VOL300’)

Example 5: This example will add one year to the date with format YYMMDD in the 
specified position if a record contains ‘APP1’ or ‘APP3’ in the first 9 positions. Only 
these selected records are written to the output data set.

DD01 COPYREC RDW=3,MAXDATERR=100,PIVOTYR=70,
IF=(1,9,C’APP1’),
    OR=(1,9,C’APP3’),
WARP=(10,C,YYMMDD,ADD=1Y)

Example 6: This example will copy records containing an 8 or 9 and will warp each 
record with YYDDD or YYMMDD data format. If a record has both formats, only the 
YYDDD format is warped.

DD01 COPYREC IF=(12,EQ,C’8’),
  WARP=(16,C,YYDDD,ADD=5Y),
  IF=(22,EQ,C’9’),
  WARP=(26,C,YYMMDD,ADD=5Y)

Example 7: This example will copy all records and warp all 8 and 9 dates.

DD01 COPYALL IF=(12,EQ,C’8’),
WARP=(15,C,YYDDD,ADD=5Y),
IF=(22,EQ,C’9’),
WARP=(26,C,YYMMDD,ADD=5Y)                             

Example 8: This example will copy records containing an 8 or 9 and warp all dates on 
these records.

DD01 COPYSOME IF=(12,EQ,C’8’),
 WARP=(16,C,YYDDD,ADD=5Y),
 IF=(22,EQ,C’9’),
 WARP=(26,C,YYMMDD,ADD=5Y)

Example 9: Any record containing a ‘111’ anywhere in the record will be replaced with 
‘ZZZ’. Any record containing a ‘DDD’ in column 1 will be replaced with a ‘---‘. Any 
record containing ‘ZZZZ’ in column 1 will not be written to the output file.

DD01 EXCLUDEREC OVERALL=(1,0,C’111’, C’ZZZ’),
   OVERLAY=(1,EQ,C’DDD’,C’---’),
   IF=(1,EQ,C’ZZZZ’),MAXRECOUT=0

Example 10: This example will select every fifth record, and, if it contains a ‘241104’ 
or a ‘181022’ in position 25, will output the record. A maximum of 5 records can be 
written to the output data set identified by DD5.

DD01 MULTICOPY SELECT=5,IF=(25,EQ,C’241104,181022’),WRITE=DD5,MAXRECOUT=5  

Example 11: This example will write all records having ‘011’ in column 9 and ‘01’ in 
column 1 to the data set identified by DD OUT1. It will write all records having ‘011’ in 
column 9 and not having a “U” or a “V” in column 1 to the data set identified by DD 
OUT2. It will write all records not having a ‘011’ in column 9 to OUT2.

DD01 MULTICOPY IF=(9,EQ,C’011’),IF=(1,EQ,C’01’),WRITE=OUT1,
      IF=(9,EQ,C’011’),IF=(1,NE,X’E4,E7’),WRITE=OUT2,
      IF=(9,NE,C’011’),WRITE=OUT2  

Example 12: This example will insert a JOBLIB DD statement after the jobcard in all 
members prefixed with PRS. These members will be written to a new data set 
identified in the JCL by DD NEWJCL. 

DD01 MULTICOPY MEMBERS=PRS,OPTIONS=JCL,WRITE=NEWJCL,
  IF=(3,0,C’ JOB '),MOVE=(1,80C' '),
  MOVE=(1,C’//JOBLIB DD DISP=SHR’),
  MOVE=(+0,C’,DSN=JOBLIB.DSN’),WRITE=NEWJCL

Example 13: This example will process three input data sets and write the results to a 
single output file identified by DD DD02O. Control statement DD02 selects a member 
called C123.

DD01 MULTICOPY WRITE=DD02O
DD02 COPYMBR MEMBER=C123
DD03 MULTICOPY WRITE=DD02O

Example 14: If two blanks are found in column 1, they will be changed to ‘**’. 
Changed records will be written to the output file identified by OUTPUTA in the JCL. 
Processing will stop when ‘END’ is found in column 1.

DD01 MULTICOPY CHANGE=(1,EQ,C’  
’,C’**’),STOPIF=(1,EQ,C’END’),WRITE=OUTPUTA

Example 15: This example will write ‘CATALOG NAME=’ to the output buffer followed 
by the 15 bytes 11 relative positions after the ‘I’ in ‘IN-CAT’ when ’IN-CAT’ is found 
anywhere in the record. It will write the 15 bytes to the current relative position in the 
output buffer. Then, it will write the record to the output data set specified by 
CATREPT.

DD01 MULTICOPY PADCHAR=C’ ’,
      IF=(1,0,C’IN-CAT’),

MOVE=(1,C’CATALOG NAME=’),
MOVE=(+0,15,+11),
WRITE=CATREPT,MAXRECOUT=0    

Example 16: This example will replace the first occurrence of text string ‘CYCLE’ with 
blanks and will print the record if text string ‘** AAAA’ is found anywhere in a record.

DD01 PRINTCHR IF=(1,0,T’** AAAA’),
    OVERLAY=(1,0,T’cycle’,5C’ ’),MAXRECOUT=0   

Example 17: This example will list the contents of all members in the partitioned data 
set that match the member mask specified. For example, members that match the 
member mask include TEST111 and TESTAB1.

DD01 PRINTCHRMBR MEMBERS=TEST--1  

Example 18: This example will process the input data set twice, once for each 
PRINTHEX control statement. Records containing ‘EDIT’ or ‘A99’ in position 19 will be 
printed in hexadecimal format. 

DD01 PRINTHEX OPTIONS=MULTI,IF=(19,EQ,C’EDIT’)
DD01 PRINTHEX IF=(19,EQ,C’A99’)

Example 19: This example will change the ‘SYS001’ to a ‘SYSTST’ for members 
containing a ‘SYS001’ anywhere in the first 20 positions of the record.

DD01 UPDATEMBR CHANGE=(1,20,’SYS001’,C’SYSTST’)

Example 20: This example will change all occurrences of hex ‘98365C’ to hex 
‘99365C’ if a hex ‘98365C’ is found anywhere in the record. The second IF parameter 
statement is logically OR’d with the first. In the second IF parameter, if a hex ‘98000C’ 
is found anywhere in the record, all occurrences of hex ‘98000C’ will be changed to 
hex ‘99000C’. 

DD01 UPDATEREC IF=(1,0,X’98365C’),OVERALL=(1,0,X’98365C’,X’98365C’)
       IF=(1,0,X’98000C’),OVERALL=(1,0,X’98000C’,X’99000C’)

Example 21: This example will update the data set in place and replace all 
occurrences of a ‘!’ or a ‘:’ or a ‘?’ with a blank.

DD01 UPDATEREC REPLALL=(1,0,C’!,:,?’,C’ ’) 

Example 22: This example will skip records 1 though 10, output records 11 through 
20, skip records 21 through 25, then output records 26 through 45.

DD01 SKIP MAXRECIN=10
DD01 COPYREC MAXRECIN=10,MAXRECOUT=10
DD01 SKIP MAXRECIN=5
DD01 COPYREC MAXRECIN=20,MAXRECOUT=20

Example 23: This example will skip the last two records beginning at the end of a 
sequential data set. Any remaining records that have ‘01SER20’ in column 1 will be 
printed in reverse order.

DD01 SKIPREV MAXRECIN=2
DD01 PRINTREV IF=(1,EQ,C’01SER20’)   

The following example illustrates several StarBat facilities.

StarBat JCL and Control Facilities 

//SAMPLEA  JOB ...
//PRINT1 EXEC PGM=StarBat,TIME=1,REGION=2M
//SYSPRINT DD SYSOUT=(,)
//SYSLIST  DD SYSOUT=(,)
//SYSTOTAL DD SYSOUT=(,)
//DD01      DD DSN=SAMPLE.LIB.PDSE(StarBatT),DISP=SHR
//DD01O     DD DSN=SAMPLE.LIB.ASM(StarBatT),DISP=SHR
//SYSIN     DD *
DD01 COPYSOME  IF=(1,EQ,C'1800'),
            WARP=(6,C,YYDDD,ADD=5Y),PRINTHEX=1,
            SUM=(26,4,C,C'The 1800 totals'),
            IF=(12,EQ,C'1900'),
            WARP=(17,C,YY,ADD=2Y),PRINTHEX=1,
            SUM=(42,3,C)
//

Sample SYSPRINT Output

PDS100I StarBat/Both -- Version 7.4.0  2000.350

Proprietary software product of SERENA Software
Phone (650)696-1800 Support 696-1850 startool@serena.com
LICENSED TO: StarTool will not work after Dec 31, 1998
             To extend, contact SERENA at (650)696-1800
All other rights reserved - use of this software
product by unauthorized persons is strictly prohibited.

DD01 COPYSOME  IF=(1,EQ,C'1800'),
            WARP=(6,C,YYDDD,ADD=5Y),PRINTHEX=1,

            SUM=(26,4,C,C'The 1800 totals'),
            IF=(12,EQ,C'1900'),
            WARP=(17,C,YY,ADD=2Y),PRINTHEX=1,
            SUM=(42,3,C)
 *** End of control statement
 
PDS220I //DD01      DD  DSN=SAMPLE.LIB.PDSE,DISP=SHR,UNIT=3380,
PDS220I //  DCB=(RECFM=FB,LRECL=80,BLKSIZE=11440,DSORG=PX),VOL=SER=SER003,
PDS220I //  SPACE=(CYL,(420,20))                 /*FREE TRK=1029*/

STRB01I BSAM input is in use
STRB05I DDNAME=SYS00001 DSN=SAMPLE.LIB.ASM opened for BSAM output
STRB05I DCB=(RECFM=FB,LRECL=80,BLKSIZE=11440),VOL=SER=SER003
--- STRB75E WARP= Invalid character number at column 0006; 
hex=X'F6F9F1C1F3';
                                              char=C'691A3'; at record 7
Text:             WARP=(6,C,YYDDD,ADD=5Y)

-- STRB80E SUM= invalid numeric character; char=C'00X0'; at record 7
TEXT:             SUM=(26,4,C,C'The 1800 totals')

Actions taken for:       IF=(1,EQ,C'1800')               --------7 --------4
Actions taken for:  WARP=(6,C,YYDDD,ADD=5Y)              --------4 --------3
Actions taken for:  SUM=(26,4,C,C'The 1800 totals')      --------4 -------
-3
Actions taken for:  IF=(12,EQ,C'1900')                   --------6 --------3
Actions taken for:  WARP=(17,C,YY,ADD=2Y)                --------3 --------3
Actions taken for:  SUM=(42,3,C)                         --------3 --------3

 Sample SYSLIST Output

StarBat Version 7.4.0;  Jan  5, 2001;  DD01=SAMPLE.LIB.PDSE(StarBatT)        VOL=SER003  PAGE   1

RECORD      2 DATA    80   CHAR 1800+74123-1900-98-2----+0034+----+----4-123+----5...7----+----8
   ***C H A N G E D***     ZONE FFFF4FFFFF6FFFF6FF6F66664FFFF4666646666F6FFF46666F666F666646666F
                           NUMR 1800E7412301900098020000E0034E0000E000040123E0000500070000E00008
                                ----+----1----+----2----+----3----+----4----+----5...7----+----8

RECORD      2 DATA    80   CHAR 1800+74123-1900-00-2----+0034+----+----4-123+----5...7----+----8
   ***C H A N G E D***     ZONE FFFF4FFFFF6FFFF6FF6F66664FFFF4666646666F6FFF46666F666F666646666F
                           NUMR 1800E7412301900000020000E0034E0000E000040123E0000500070000E00008
                                ----+----1----+----2----+----3----+----4----+----5...7----+----8

RECORD      7 DATA    80   CHAR 1800+691A3-1900-A8-2----+00X0+-*ERROR--4-00X+----5...7----+----8
   *****E R R O R*****     ZONE FFFF4FFFCF6FFFF6CF6F66664FFEF465CDDDD66F6FFE46666F666F666646666F
                           NUMR 1800E6911301900018020000E0070E0C599690040007E0000500070000E00008
                                ----+----1----+----2----+----3----+----4----+----5...7----+----8

RECORD      7 DATA    80   CHAR 1800+691A3-1900-A8-2----+00X0+-*ERROR--4-00X+----5...7----+----8
   *****E R R O R*****     ZONE FFFF4FFFCF6FFFF6CF6F66664FFEF465CDDDD66F6FFE46666F666F666646666F
                           NUMR 1800E6911301900018020000E0070E0C599690040007E0000500070000E00008
                                ----+----1----+----2----+----3----+----4----+----5...7----+----8

Sample SYSTOTAL Output

SYSTOTAL Report

                                 Following totals developed from
                                 SAMPLE.LIB.PDSE  VOL=SER003

'The 1800 totals                                   ----------            155
            SUM=(42,3,C)                           ----------            458

StarBat issues the following return codes to indicate the results of job execution:

StarBat issues many STRBnnx and PDSnnnx format messages that can be found in 
the Serena™ StarTool® FDM Messages Guide.  

The following unnumbered messages are unique to StarBat and are documented 
here.

UNNUMBERED MESSAGES

*** End of control statement
The last parameter on a function has been input; the function can now be executed.

*** Last of control statements
The last input record has been input in the control statement data set; the end of file 
was reached.

Return Code Description

0 All functions completed successfully

4 An error was found in decoding a control card; all control 
statement processing is terminated.

8 No user records were copied to an output data set during a 
COPYREC, COPYALL, COPYMBR, COPYREV, 
EXCLUDEREC or MULTICOPY function.

12 An input or output master file could not be opened; StarBat 
terminates all processing.

*** EXCLUDEREC= parameter has been satisfied
The EXCLUDEREC parameter has been satisfied; this EXCLUDEREC function is 
complete.

*** MAXRECIN= parameter satisfied
The MAXRECIN parameter has been satisfied; this function is complete.

*** MAXRECOUT= parameter has been satisfied
The MAXRECOUT parameter has been satisfied; this function is complete.

*** STOPIF= parameter has been satisfied
The STOPIF parameter has been satisfied; this function is complete

*** MAXDATERR= parameter has been satisfied
Too many invalid dates have been encountered; this function is complete

*** No //SYSIN DD statement
No SYSIN DD statement was entered; all corresponding //DDxx data sets will be 
copied to //DDxxO data sets.

*** No //SYSIN DD controls
The SYSIN DD data set contained no StarBat control statements; all corresponding /
/DDxx data sets will be copied to //DDxxO data sets.

DDxx COPYREC   (generated statement)
Matching //DDxx and //DDxxO data sets have been found and the input data set is 
being copied to the output data set.

